Rev 1 - use alternate solution of psi1, psi2.



Theta6 = 100deg and Theta7 = 20deg
Link 1 is 9m long, Link 2 is 5m long, Link 3 is 7m long, Link 4 is 9m long
Lamda2 = 50deg
lamda3 = 55deg
lamda4 = 50deg
lamda5 = -20 deg

Initialization, Constants, Procedures:

```
reset()
```

```
use(plot):
Warning: 'hull' already has a value, not exported. [use]
Warning: 'Integral' seems to be proteced; not exported. [use]
Warning: 'Pyramid' seems to be proteced; not exported. [use]
```

```
iexp:=proc(theta) begin exp(I*theta) end_proc:
```

```
arr:=proc(v1,v2) begin plot::Arrow2d([Re(v1),Im(v1)],
               [Re(v2),Im(v2)],Color=mycolor,
               LineWidth=mylinewidth)
                            end_proc:
```

```
lin:=proc(v1,v2) begin plot::Line2d([Re(v1),Im(v1)],
                  [Re(v2),Im(v2)],Color=mycolor,
                   LineWidth=mylinewidth)
                            end_proc:
```

```
dtr:=float(PI/180):
```

```
L1:=9: L2:=5: L3:=7: L4:=9:
```

```
lambda2  := 50*dtr:
lambda3  := 55*dtr:
lambda4  := 50*dtr:
lambda5  := -20*dtr:
theta6:=100*dtr:
theta7:=20*dtr:
```

```
T0:=3 // Value selected for plotting convenience (all vectors scale)
```

3

Solve to find angles Phi1 and Phi2:

```
EqPhi:=L1*iexp(theta6)+L2*iexp(theta6+phi1)=
    L4*iexp(theta7)+L3*iexp(theta7+phi2)
```

$$5\,e^{\mathrm{phi1}\,i+1.745329252\,i}-1.562833599+8.863269777\,i=7\,e^{\mathrm{phi2}\,i+0.3490658504\,i}+8.457233587+3.07818129\,i$$

```
assume (phi1 in R_)
```

```
assume(phi2 in R_)
```

```
EqPhi1:=Re(lhs(EqPhi))=Re(rhs(EqPhi))
```

$$5.0\cos(phi1+1.745329252)-1.562833599=7.0\cos(phi2+0.3490658504)+8.457233587$$

```
EqPhi2:=Im(lhs(EqPhi))=Im(rhs(EqPhi))
```

$$5.0\sin(phi1+1.745329252)+8.863269777=7.0\sin(phi2+0.3490658504)+3.07818129$$

```
PhiSoln:=numeric::solve({EqPhi1,EqPhi2},[phi1=2,phi2=1])
```

$$\{[phi1=-1.950050314,\ phi2=2.043083707]\}$$

```
phi1:=phi1|PhiSoln
```

$$-1.950050314$$

```
phi2:=phi2|PhiSoln
```

$$2.043083707$$

```
phi1/dtr
```

$$-111.7296528$$

```
phi2/dtr
```

$$117.0600736$$

Solve to find tension T1 and T2 to balance forces applied at end of bar 1

```
Eq1:=T0*iexp(PI-lambda5)+T0*iexp(lambda4)+T1*iexp(theta6-PI)
  +T2*iexp(theta6+phi1)=0
```

$$T1\,(0.1736481777-0.984807753\,i)+T2\,(0.9791177291-0.2032940543\,i)-0.8907150333+1.272072899\,i=0$$

```
Eq1r:=Re(lhs(Eq1))=0
```

$$\Re(T2\,(0.9791177291-0.2032940543\,i))+\Re(T1\,(0.1736481777-0.984807753\,i))-0.8907150333=0$$

```
Eq1i:=Im(lhs(Eq1))=0
```

$$0.1736481777\,\Im(T1)+0.9791177291\,\Im(T2)-0.984807753\,\Re(T1)-0.2032940543\,\Re(T2)+1.272072899=0$$

```
T12soln:=solve({Eq1r,Eq1i},{T1,T2})
```

$$\{[T1=1.145855287,\ T2=0.7064925189]\}$$

```
T1:=T1|T12soln
```

$$1.145855287$$

```
T2:=T2|T12soln
```

$$0.7064925189$$

Solve to find tension T3 and T4 to balance forces applied at end of bar 4

```
Eq4:=T0*iexp(PI/2-lambda3)+T0*iexp(-PI/2+lambda2)
+T4*iexp(theta7-PI)+T3*iexp(theta7+phi2)=0
```

$$T3\,(-0.7320683629+0.6812311737\,i)+T4\,(-0.9396926208-0.3420201433\,i)+\frac{3}{e^{\frac{\pi i}{2}-0.872664626\,i}}+3\,e^{\frac{\pi i}{2}-0.9599310886\,i}=0$$

```
Eq4r:=Re(lhs(Eq4))=0
```

$$\Re(T4\,(-0.9396926208-0.3420201433\,i))+3.0\cos\left(0.872664626-\frac{\pi}{2}\right)+3.0\cos\left(\frac{\pi}{2}-0.9599310886\right)+\Re(T3\,(-0.7320683629+0.6812311737\,i))=0$$

```
Eq4i:=Im(lhs(Eq4))=0
```

$$3.0\sin\left(\frac{\pi}{2}-0.9599310886\right)-0.7320683629\,\Im(T3)-0.9396926208\,\Im(T4)+0.6812311737\,\Re(T3)-0.3420201433\,\Re(T4)+3.0\sin\left(0.872664626-\frac{\pi}{2}\right)=0$$

```
T34soln:=solve({Eq4r,Eq4i},{T3,T4})
```

$$\{[T3=2.045544797,\ T4=3.467209127]\}$$

```
T3:=T3|T34soln
```

$$2.045544797$$

```
T4:=T4|T34soln
```

$$3.467209127$$

Solve to find Spring force vector for info

```
EqS:=T2*iexp(theta6+phi1-PI)+T3*iexp(theta7+phi2-PI)+Ts*iexp(thetaS-PI)=0
```

$0.8057392798 - Ts\, e^{thetaS\, i} - 1.249863154\, i = 0$

```
EqSr:=Re(lhs(EqS))=0
```

$0.8057392798 - \Re(Ts\, e^{thetaS\, i}) = 0$

```
EqSi:=Im(lhs(EqS))=0
```

$-\Im(Ts\, e^{thetaS\, i}) - 1.249863154 = 0$

```
Tssoln:=solve({EqSr,EqSi},{Ts,thetaS})
```

$\{[Ts = 1.487068825,\ thetaS = -0.9981828126]\}$

First generate the bars: B1 thru B4

```
mylinewidth:=2:
mycolor:=RGB::Blue:

B1:=lin(0,L1*iexp(theta6)):
B4:=lin(0,L4*iexp(theta7)):
B2:=lin(L1*iexp(theta6),L1*iexp(theta6)+L2*iexp(theta6+phi1)):
B3:=lin(L4*iexp(theta7),L4*iexp(theta7)+L3*iexp(theta7+phi2)):
```

Now generate the applied forces of magnitude T0

```
mylinewidth:=1:
mycolor:=RGB::Red:
AT04:=arr(L1*iexp(theta6),L1*iexp(theta6)+T0*iexp(lambda4)):

AT05:=arr(L1*iexp(theta6),L1*iexp(theta6)+T0*iexp(PI-lambda5)):
AT0405:=arr(L1*iexp(theta6),L1*iexp(theta6)+T0*iexp(PI-lambda5)
  +T0*iexp(lambda4)):
AT02:=arr(L4*iexp(theta7),L1*iexp(theta7)+T0*iexp(-PI/2+lambda2)):
AT03:=arr(L4*iexp(theta7),L1*iexp(theta7)+T0*iexp(PI/2-lambda3)):
AT0304:=arr(L4*iexp(theta7),L1*iexp(theta7)
   +T0*iexp(PI/2-lambda3)+T0*iexp(-PI/2+lambda2)):

```

Now generate the vectors for tension in the bars

```
mycolor:=RGB::Green:

AT1:=arr(L1*iexp(theta6),L1*iexp(theta6)+T1*iexp(theta6-PI)):
AT2:=arr(L1*iexp(theta6),L1*iexp(theta6)+T2*iexp(theta6+phi1)):
AT3:=arr(L4*iexp(theta7),L4*iexp(theta7)+T3*iexp(theta7+phi2)):
AT4:=arr(L4*iexp(theta7),L4*iexp(theta7)+T4*iexp(theta7-PI)):
AT2A:=arr(L1*iexp(theta6)+L2*iexp(theta6+phi1),
L1*iexp(theta6)+L2*iexp(theta6+phi1)+T2*iexp(theta6+phi1-PI)):

AT3A:=arr(L4*iexp(theta7)+L3*iexp(theta7+phi2),
L4*iexp(theta7)+L3*iexp(theta7+phi2)+T3*iexp(theta7+phi2-PI)):
AT34:=arr(L4*iexp(theta7),L4*iexp(theta7)
   +T3*iexp(theta7+phi2)+T4*iexp(theta7-PI)):
AT12:=arr(L1*iexp(theta6),L1*iexp(theta6)
   +T1*iexp(theta6-PI)  +T2*iexp(theta6+phi1)):
ATS:=arr(L1*iexp(theta6)+L2*iexp(theta6+phi1),
 L1*iexp(theta6)+L2*iexp(theta6+phi1)-T2*iexp(theta6+phi1-PI)
  -T3*iexp(theta7+phi2-PI)):
ATSOpp:=arr(L1*iexp(theta6)+L2*iexp(theta6+phi1),
 L1*iexp(theta6)+L2*iexp(theta6+phi1)+T2*iexp(theta6+phi1-PI)
   +T3*iexp(theta7+phi2-PI)):
mycolor:=RGB::Violet:
ATS:=arr(L1*iexp(theta6)+L2*iexp(theta6+phi1),
 L1*iexp(theta6)+L2*iexp(theta6+phi1)-T2*iexp(theta6+phi1-PI)
    -T3*iexp(theta7+phi2-PI)):
plot([B1,B2,B3,B4,AT02,AT03,AT04,AT05,AT0304,AT0405,
AT1,AT2,AT3,AT4,AT12, AT34, AT2A,AT3A,ATS,ATSOpp],
Scaling=Constrained,ViewingBoxXRange =-3..12,
                       XGridVisible,YGridVisible)
```



There are two solutions of (psi1,psi2). The first one posted previously. In this 2nd one as in the first, the purple vector (spring force) is not in line with the spring.